

Mathematical Programming: Modelling and Applications

October 2009

Sonia Cafieri

LIX, École Polytechnique

caferi@lix.polytechnique.fr



AMPL model – Dyeing

```
param l >= 1;  param v >= 1;

set L := 1..l;
set V := 1..v;
set V0 := 1..v-1;

param s{L,V} >= 0;
param M default sum{i in L, k in V} s[i,k] ;

var t{L,V} >= 0;
var T >= 0;
var y{L,L,V} binary;

minimize makespan : T;

subject to sequential{i in L, k in V0} : t[i,k] + s[i,k] <= t[i,k+1];

subject to lastbath{i in L} : t[i,v] + s[i,v] <= T;

subject to nonoverlap{i in L, j in L, k in V : i != j} :
    t[i,k] + s[i,k] <= t[j,k] + M * (1 - y[i,j,k]);

subject to disjunction{i in L, j in L, k in V : i != j} :
    y[i,j,k] + y[j,i,k] = 1;
```



AMPL dat – Dyeing

```
param l := 5;
```

```
param v := 3;
```

```
param s : 1 2 3 :=
```

```
1      3.0 1.0 1.0
```

```
2      2.0 1.5 1.0
```

```
3      3.0 1.2 1.3
```

```
4      2.0 2.0 2.0
```

```
5      2.1 2.0 3.0
```

```
;
```



AMPL run – Dyeing

```
model dyeing.mod;
data dyeing.dat;

option solver cplex;
solve;

display makespan;

for {i in L} {
  printf "batch %d : ", i;
  for {k in V} {
    printf "[%f] ", t[i,k];
  }
  printf "\n";
}
```



Solution – Dyeing

```
ILOG AMPL 10.100, licensed to "ecolepolytechnique-palaiseau".  
AMPL Version 20060626 (Linux 2.6.9-5.ELsmp)  
ILOG CPLEX 10.100, licensed to "ecolepolytechnique-palaiseau", options: e  
m b q use=8  
CPLEX 10.1.0: optimal integer solution; objective 14.1  
1175 MIP simplex iterations  
235 branch-and-bound nodes  
makespan = 14.1
```

```
batch 1 : [9.100000] [12.100000] [13.100000]  
batch 2 : [0.000000] [2.000000] [3.500000]  
batch 3 : [6.100000] [10.600000] [11.800000]  
batch 4 : [2.000000] [4.000000] [6.000000]  
batch 5 : [4.000000] [6.100000] [8.100000]
```